



Don't Be a SAS® Dinosaur: Modernize Your SAS Programs

by Warren Repole


Variable Names: Rename a Numbered Range List

Scenario:

You have a set of variables whose names start with the same "root" characters and end with a number in sequence. You wish to change the "root" part of each name and/or the numeric "suffix" values.

The traditional rename syntax has each individual variable listed with its corresponding new name.

The old way: Macro-Generated Rename List



The original approach is to build a rename list using a macro loop.

```
%macro renameloop(oldprefix=X,newprefix=Y,
                 oldstart=1,newstart=1,count=1);
  %local i;
  %do i=1 %to &count;
    &oldprefix%eval(&oldstart+&i-1) = &newprefix%eval(&newstart+&i-1)
  %end;
%mend renameloop;
proc sort data=sashelp.pricedata out=pricedata nodupkey;
  by Date;
run;
%let renames=%renameloop(oldprefix=Price,newprefix=UnitPriceProduct,count=17);
options symbolgen;
data Prices2002;
  set pricedata;
  where Date between "01jan2002"d and "30jun2002"d;
  rename &renames ;
run;
options nosymbolgen;
proc print data=Prices2002 heading=h;
  id Date;
  format Date date9.;
  var UnitPrice;;
  title "Product Prices throughout First Half of 2002";
run;
```

Please direct all correspondence to:

Warren Repole, 1705 Palm Springs Dr, Vienna VA 22182-2331

sasdinosaur@repole.com

www.repole.com/dinosaur

The new way: Use a Variable List as the Rename List (available in SAS Version 8)

An alternate approach is to use the numbered range list form of the RENAME= data set option and/or RENAME statement.

```
proc sort data=sashelp.pricedata out=pricedata nodupkey;
  by Date;
run;
data Prices2002;
  set pricedata;
  where Date between "01jan2002"d and "30jun2002"d;
  rename Pricel-Price17 = UnitPriceProduct01-UnitPriceProduct17;
run;
proc print data=Prices2002 heading=h;
  id Date;
  format Date date9.;
  var UnitPrice:;
  title1 "Product Prices throughout First Half of 2002";
run;
```

Advantages of the alternate approach:

- No macro code is involved, especially indirect macro variable references.
- It is not necessary to list each individual variable within the rename list syntax.
- Numeric suffixes can be shifted easily.

Disadvantages of the alternate approach:

- None identified

Additional documentation for this technique can be found in *SAS® 9.2 Language Reference: Dictionary*. Cary, NC: SAS Institute Inc.

Visit <http://support.sas.com/documentation/onlinedoc/sas9doc.html> for SAS 9 documentation.

Go to <http://www.repole.com/dinosaur/renamelist.html> for the sample code and output for this topic.

These techniques are mentioned in other SAS references and publications:

SAS Sample 26009 (<http://support.sas.com/kb/26/009.html>)

SUGI Paper 029-30 (<http://www2.sas.com/proceedings/sugi30/029-30.pdf>)